

SCIENCE & VIE

MENSUEL

N° 833 FÉVRIER 1987

L'ANESTHÉSIE EN QUESTION

ISSN 0036 8369
BELGIQUE 110 FB - ESPAGNE 475 Ptas - MAROC 17 Dh - TUNISIE 2,09 DT

SUISSE 5 FS - CANADA \$2,95 -

LA PILULE QUI SUPPRIME
L'IVRESSE

M 2578 - 833 - 16,00 F



3792578016008

08330

Arrosez votre champignon

INFORMATIQUE AMUSANTE

Ce mois-ci, nous proposons un jeu mycologique. Passionné de champignons vous désirez en cultiver un : munis d'un pot, vous voulez y faire pousser une souche. Mais cela demande de l'eau... et la sécheresse menace. Seule solution donc, récupérer les gouttes qui tombent sur la champignonnière.

Cependant cette opération n'est pas si simple. En effet les gouttes rebondissent sur le chapeau des champignons et ne tombent pas toujours au même endroit. Le but sera d'arriver à arroser suffisamment la terre du pot pour que la spore parvienne à pousser.

Divers niveaux de difficulté sont à votre disposition ainsi qu'un dispositif d'assistance si vous débutez. Notons — si le côté probabilité du jeu vous intéresse —, que la répartition des gouttes en bas de l'écran correspond à celle du diagramme gaussien ; la probabilité qu'une goutte rebondisse vers la droite ou vers la gauche après avoir frappé un chapeau étant la même.

Afin de profiter pleinement de la palette de couleurs de l'Amstrad, nous le ferons travailler en mode 0. La ligne 10 du programme fixera donc ce mode de fonctionnement et, par la même occasion, les couleurs du bord de l'écran et du fond. Diverses illustrations sonores (lignes 50 à 80) suivront le déroulement du jeu.

L'ordinateur demandera le niveau de jeu souhaité et contrôlera s'il fait partie du choix proposé (lignes 120 à 160) puis la redéfinition des graphiques nécessaires à la représentation des champignons et du pot sera effectuée des lignes 210 à 250.

La mise en place des champignons sur l'écran sera effectuée à l'aide de deux boucles imbriquées, formées autour de L et H, chargées respectivement de les positionner horizontalement et verticalement (lignes 310 à 360). Les diverses variables du jeu seront alors initia-

lisées et ce dernier pourra commencer.

Pour cela la chute de la goutte d'eau s'effectuera selon trois phases. La première concerne le haut de l'écran et sera réalisée par les lignes 470 à 510. La seconde correspond aux déplacements de la goutte parmi les champignons et fait essentiellement appel à la sous-routine 1020. Enfin, la troisième gère la fin de la chute de la goutte en bas de l'écran. Placée des lignes 600 à 660 elle fera éventuellement appel à la sous-routine d'assistance 1280 si telle a été l'option de jeu choisie. Vient alors le contrôle du résultat ; à savoir, si la goutte est tombée ou non dans le pot.

Pour cela les lignes 700 à 720 compareront les positions respectives du pot et de la goutte et ajouteront éventuellement un point à la variable CG ; variable affectée au décompte des coups réussis. De même, le crédit de temps sera amputé d'un point et la ligne 760 contrôlera que le temps de jeu imparti n'est pas épuisé. S'il l'est, le nombre de gouttes recueillies sera indiqué et une pression sur une touche commandera le début d'une nouvelle partie. Notons que le programme contrôlera également si un nombre de gouttes suffisant à la pousse du champignon a été recueilli. Les lignes 850 à 980 se chargeront de cette opération. Si le but est atteint, le champignon poussera dans le pot et le score sera affiché, il faudra taper ESPACE pour entamer une nouvelle partie. Tant qu'il restera du temps et que le champignon n'aura pas poussé, le jeu sera rebouclé.

Photo M. Toscas/Galerie 27



```

10 MODE 0:PAPER 9:BORDER 3:CLS
20 REM *****
30 REM * ENVELOPPE DES SONS. *
40 REM *****
50 ENV 2,1,15,1,1,0,2,15,-1,10
60 ENV 1,1,14,1,1,0,1,14,-1,1
70 ENV 3,1,10,1,1,0,1,5,-2,1
80 ENT 1,1,0,1,14,5,1,1,0,1
90 REM *****
100 REM * CHOIX DU NIVEAU *
110 REM *****
120 LOCATE 1,5:PRINT "CHOISISSEZ DE 1 A 6"
130 LOCATE 1,10:INPUT "VOTRE NIVEAU";NI
140 LOCATE 1,15:INPUT "ASSISTANCE (o-n)";D$
150 IF NI<1 OR NI>6 THEN GOTO 130
160 CLS
170 REM *****
180 REM * SYMBOLES GRAPHIQUES *
190 REM *****
200 SYMBOL AFTER 127
210 SYMBOL 128,24,60,126,126,255,255,255,126
220 SYMBOL 129,24,24,24,60,60,60,60,24
230 SYMBOL 130,16,16,24,56,56,56,16
240 SYMBOL 132,0,0,0,60,102,66,102,126
250 SYMBOL 133,60,24,36,60,60,60,60,24
260 LOCATE 10,23:PRINT CHR$(132)
270 LOCATE 10,24:PRINT CHR$(133)
280 REM *****
290 REM * MISE EN PLACE DES CHAMPIGNONS *
300 REM *****
310 FOR L=10 TO 18 STEP 2
320 FOR H=20-L TO L STEP 4
330 LOCATE H,L:PEN 3:PRINT CHR$(128)
340 LOCATE H,L+1:PEN 1:PRINT CHR$(129)
350 NEXT H
360 NEXT L
370 REM *****
380 REM * INITIALISATION DES VARIABLES *
390 REM *****
400 LET XJ=10:LET XJP=1:LET X=10
410 LET CG=0:LET CR=60
420 PEN 5:LOCATE 11,2:PRINT "Temps:";CR
430 PEN 6
440 REM *****
450 REM * PREMIERE CHUTE DE LA GOUTTE *
460 REM *****
470 FOR Y=2 TO 9
480 LOCATE 10,Y-1:PRINT " "
490 LOCATE 10,Y:PRINT CHR$(130)
500 GOSUB 1150
510 NEXT Y
520 LET X=10
530 LET Y=9
540 FOR C=1 TO 5
550 GOSUB 1020
560 NEXT C
570 REM *****
580 REM * FIN DE CHUTE DE LA GOUTTE *
590 REM *****
600 FOR C=19 TO 22
610 GOSUB 1150
620 IF D$="o" THEN GOSUB 1280
630 LOCATE X,C-1:PRINT " "
640 LOCATE X,C:PRINT CHR$(130)
650 NEXT C
660 LOCATE X,22:PRINT " "
670 REM *****
680 REM * CONTROLE GOUTTE DANS LE POT *
690 REM *****

```

```

700 IF XJ=X THEN LET CG=CG+1: SOUND 1,60,0,0,1
,1,0:SOUND 2,24,0,0,1,1,0
710 LET CR=CR-1
720 PEN 5:LOCATE 17,2:PRINT CR:PEN 6
730 REM *****
740 REM * CONTROLE TEMPS DE JEU ECOULE *
750 REM *****
760 IF CR>0 THEN GOTO 850
770 SOUND 1,700,0,0,2,0,0
780 SOUND 2,705,0,0,2,0,0
790 CLS
800 LOCATE 1,10:PRINT "TEMPS DE JEU ECOULE"
810 LOCATE 1,15:PRINT CG;"DANS LE POT!"
820 LOCATE 1,20:PRINT "TAPER 'ESPACE' S.V.P"
830 IF INKEY$<>" " THEN GOTO 830
840 GOTO 10
850 IF CG<>(10*NI) THEN GOTO 980
860 REM *****
870 REM * CONTROLE PARTIE GAGNEE *
880 REM *****
890 PEN 7:LOCATE XJ,21:PRINT CHR$(128)
900 PEN 13:LOCATE XJ,22:PRINT CHR$(129)
910 FOR I=20 TO 1 STEP -1
920 SOUND 3,0,1,7,0,0,1
930 NEXT I
940 LOCATE 1,2:PRINT "BRAVO;SCORE:";CR*NI*10;"
"
950 LOCATE 1,4:PRINT "TAPER 'ESPACE' S.V.P"
960 IF INKEY$<>" " THEN GOTO 960
970 GOTO 10
980 GOTO 470
990 REM *****
1000 REM * REBONDISSEMENT DE LA GOUTTE *
1010 REM *****
1020 LET D=1-2*INT(RND*2)
1030 FOR M=1 TO 4 STEP 2
1040 LOCATE X,Y:PRINT " "
1050 LET X=X+D:IF X=0 THEN LET X=1
1060 LET Y=Y+1
1070 LOCATE X,Y:PRINT CHR$(130)
1080 SOUND 3,X+Y,0,0,3,0,0
1090 GOSUB 1150
1100 NEXT M
1110 RETURN
1120 REM *****
1130 REM * DEPLACEMENT DU POT DU JOUEUR *
1140 REM *****
1150 LET XJP=XJ
1160 PEN 11
1170 LET XJ=XJ+((INKEY(1))*1-(XJ<20))-((INKEY(
8))*1-(XJ>1))
1180 IF XJ=XJP THEN FOR T=1 TO 30:NEXT T:GOTO
1230
1190 LOCATE XJP,23:PRINT " "
1200 LOCATE XJP,24:PRINT " "
1210 LOCATE XJ,23:PRINT CHR$(132)
1220 LOCATE XJ,24:PRINT CHR$(133)
1230 PEN 6
1240 RETURN
1250 REM *****
1260 REM * SOUS ROUTINE D'ASSISTANCE *
1270 REM *****
1280 LET XJP=XJ
1290 IF XJ<X THEN LET XJ=XJ+1
1300 IF XJ>X THEN LET XJ=XJ-1
1310 PEN 11
1320 IF XJP<>XJ THEN SOUND 3,0,0,0,3,1,1:GOSUB
1190
1330 PEN 6
1340 RETURN

```

Trois sous-routines, comme nous l'avons vu, sont utilisées pour ce jeu. La première, placée entre les lignes 1020 et 1110, concerne le rebondissement de la goutte sur le chapeau des champignons. La variable de direction, D, sera tirée au sort et aiguillera l'eau tout au long de sa chute. Il ne sera donc pas étonnant de la voir prise en compte pour le calcul de la position horizontale de cette dernière. La seconde, lignes 1150 à 1240, concerne le déplacement du pot. Nous utiliserons la fonction INKEY

pour contrôler sa position en fonction des flèches appuyées. Enfin la dernière est liée à la fonction assistance. Comprise entre les lignes 1280 et 1340, son but sera donc de comparer la position finale du pot et de la goutte lors de la phase finale de sa chute de façon à corriger légèrement ou non la position du pot choisie par le joueur.

La frappe de ce programme ne doit pas poser de problème. Comme toujours nous avons évité le basic abrégé.

L'utilisation est également très

simple : après avoir tapé RUN, l'ordinateur demandera d'indiquer le niveau de jeu souhaité et si une assistance est désirée. Une fois les réponses fournies à ces deux questions, le jeu débutera. Le temps imparti sera en permanence indiqué en haut à droite de l'écran et, pour déplacer le pot, il suffira d'utiliser les flèches du curseur de déplacement vers la droite ou vers la gauche. En fin de jeu, une seconde partie sera obtenue en appuyant sur la touche ESPACE.

Henri-Pierre Penel